D. MAÏTI

L'enveloppe convexe d'un ensemble de points du plan : algorithme commenté en problème ; et programme

Les cahiers de l'analyse des données, tome 4, nº 2 (1979), p. 175-188

<http://www.numdam.org/item?id=CAD_1979__4_2_175_0>

Les Cahiers de l'Analyse des données
Vol. IV - 1979 - n° 2 - p. 175-188

L'ENVELOPPE CONVEXE D'UN ENSEMBLE DE POINTS DU PLAN : ALGORITHME COMMENTÉ EN PROBLÈME ; ET PROGRAMME [CONVAP]

par D. Maïti (1)

Par cet article, débute la publication d'une suite de travaux entrepris pour l'analyse de données recueillies dans des expériences de physique corpusculaire à haute énergie. Le programme CONVAP publié ici concerne un ensemble de points du plan. Il sert de procédure auxiliaire dans un programme étudiant l'enveloppe convexe d'un ensemble de points de R^3, avec détermination des arêtes de ce convexe, et tracé de son épure (programme CONVESP) à paraître. Le programme CONVESP nous permet de déterminer pour un événement à haute énergie et haute multiplicité (collision proton-proton à 52 GeV/c dans le centre masse ; données des ISR du CERN) un ensemble de paramètres géométriques de son *hodographe* (ou ensemble des vecteurs vitesse des particules émergentes). L'analyse de ces données géométriques aboutit à des relations nouvelles qui permettent des *comparaisons* très fines entre événements physiques réels et événements simulés.

Les programmes CONVAP et CONVESP sont susceptibles d'intéresser tant les praticiens du dessin automatique, que les spécialistes des probabilités géométriques (e.g. pour l'étude par simulation des relations entre le nombre de points extrémaux d'un échantillon issu d'une loi donnée, et l'effectif total de cet échantillon). Les *comparaisons* faites dans le cadre de la physique peuvent servir ailleurs. C'est pourquoi on publie notre travail en une suite d'articles aussi indépendants que possible les uns des autres.

Le programme CONVAP est lui-même présenté ici par l'énoncé d'un problème pour l'instruction des étudiants en analyse des données qui doivent s'entraîner à la conception d'algorithmes. L'énoncé est écrit en langage ALGOL 68 ; mais nous donnons aussi le listage FORTRAN du programme utilisé sur l'ordinateur CDC 6600.

L'auteur remercie le professeur Froissart et l'équipe du laboratoire de physique corpusculaire du Collège de France, qui ont fourni les données et inspiré les recherches.

1 Enoncé du problème

L'objet du problème est de découvrir l'action d'un algorithme en le faisant jouer sur un exemple, et d'expliquer cette action en adjoignant des commentaires au programme ci-joint. Comme entrée du programme on a un ensemble de points du plan numérotés de 1 à CARD I ; on appellera simplement point I le point ayant pour abscisse XI[I] et pour ordonnée YI[I]. Après exécution du programme, CARS de ces points dont les numéros sont contenus dans le tableau ISOM se trouvent distingués. Il s'agit de découvrir ce que sont les points ISOM[1],..., ISOM[CARS] relativement à

(1) Laboratoire de physique corpusculaire, Collège de France ; et laboratoire de statistique, université Pierre et Marie Curie.

l'ensemble total des CARD I points donnés. L'étudiant est invité à lire rapidement le listage de l'algorithme (§ 2) puis à résoudre les questions dans l'ordre où elles sont posées.

Dans tout le problème on considérera l'exemple ci-dessous, où CARD I = 9 ; on lit e.g. dans les tableaux que le point I = 7 a pour abscisse XI[I] = 2 et pour ordonnée YI[I] = 0.

	1	2	3	4	5	6	7	8	9
XI	2	2	3	1	4	1	2	1	4
YI	3	2	1	3	2	2	0	1	3

1.1 On considère le bloc d'instruction :

IMI:=IMA:=1;XMI:=XMA:=XI[1];
pour I *depuis* 2 *pas* 1 *jusqu'à* CARD I *faire*
si XI[I] $<_s$ XMI *alors* XMI:=XI[I];IMI:=I
sinsi XI[I]=XMI *et* YI[I] $<_s$ YI[IMI] *alors* IMI:=I
sinsi XI[I]=XMA *et* YI[I] $>_s$ YI[IMA] *alors* IMA:=I
sinsi XI[I] $>_s$XMA *alors* XMA:=XI[I];IMA:=I *fsi fait;*

commentaire : nous utilisons les signes $<_s$ et $>_s$ pour exprimer des inégalités strictes.

a) Donner les valeurs des quatre quantités IMI, IMA, XMI, XMA, à la fin de l'exécution de chacune des 8 étapes I = 2,..., 9 de la boucle faire, dans le cas des données de l'exemple. On donnera le résultat sous forme d'un tableau ayant pour lignes l'ensemble {IMI,IMA,XMI,XMA} et pour colonnes l'ensemble des 8 étapes : I = 2, 3,..., 9. Dans ce tableau, on pourra indiquer d'une étoile les cases ayant changé de contenu par rapport à la colonne précédente.

b) Faire une figure comportant les 9 points I = 1, 2,..., 9 ; en marquant les deux points IMI et IMA obtenus à l'issue de l'exécution du bloc ci-dessus.

c) Insérer un commentaire pour expliquer ce que sont en général les points IMI et IMA déterminés par le bloc ci-dessus ; on précisera en particulier le rôle des deux instructions :

si XI[I]=XMI *et* YI[I] $<_s$ YI[IMI] *alors* IMI:=I
sinsi XI[I]=XMA *et* YI[I]$>_s$YI[IMA] *alors* IMA:=I

1.2 Imaginer un exemple de données simples avec CARD I = 4, pour lequel s'applique l'instruction :

si XMI=XMA *alors*
....
aller à FIN *fsi*

(i.e. un exemple pour lequel à l'issue du bloc étudié en 1° on ait XMI=XMA). Quel est dans ce cas l'effet de l'ensemble du programme.

1.3 On étudie maintenant l'effet de la partie du programme comprise entre l'instruction :

pour I *depuis* 1 *pas* 1 *jusqu'à* CARDI *faire* ESI[I]:=1 *fait;*

et la ligne :

CARS:=CARS+1;ISOM[CARS]:=IMA;IS:=IMA;

mais afin de simplifier cette étude, on considère d'abord le programme suivant où n'intervient pas le tableau ESI.

```
IS:=IMI;CARS:=1;ISOM[1]:=IMI;
ETIQA:
PCORD:=(YI[IMA]-YI[IS])/(XI[IMA]-XI[IS]);
MINPENTE:=PCORD;ISP:=IMA;
pour I depuis 1 pas 1 jusqu'à CARDI faire
    si XI[I]≤XI[IS] alors aller à ETIQF fsi;
    PTIIS:=(YI[I]-YI[IS])/(XI[I]-XI[IS]);
    si PTIIS<s MINPENTE alors
       MINPENTE:=PTIIS;ISP:=I
    sinsi PTIIS=MINPENTE et XI[I]>s XI[ISP] alors ISP:=I fsi
    ETIQF:fait
si XI[ISP]<s XI[IMA] alors
    CARS:=CARS+1;ISOM[CARS]:=ISP;IS:=ISP; aller à ETIQA
sinsi XI[ISP]=XI[IMA] et YI[ISP]<s YI[IMA] alors
    CARS:=CARS+1;ISOM[CARS]:=ISP fsi;
CARS:=CARS+1;ISOM[CARS]:=IMA;IS:=IMA;
```

a) Donner l'interprétation géométrique des quantités PCORD et PTIIS

b) L'instruction *faire pour* I *depuis* 1 *pas* 1 *jusqu'à* CARD I .... ETIQF: *fait*; est exécutée d'abord avec la valeur IS=IMI (où IMI a été trouvé en 1°) donner dans le cas de l'exemple les valeurs de MINPENTE et de ISP à la fin de chacune des 9 étapes I = 1,..., 9. On donnera le résultat sous forme de tableau comme pour 1°.

En déduire la valeur définitive de ISOM[2] ; marquer sur la figure le point I = ISOM[2].

c) L'instruction de boucle est exécutée une deuxième fois avec la valeur IS=ISOM[2] ; faire le même tableau qu'en b) ; expliquer d'après cet exemple quand joue l'instruction.

sinsi PTIIS=MINPENTE *et* XI[I]>$_s$XI[ISP] *alors* ISP:=I *fsi;*

Quelles sont les valeurs de ISOM[3] et ISOM[4] ? Marquer sur la figure les points I=ISOM[3] et I=ISOM[4].

1.4 On considère maintenant la partie du programme comprise entre ETIQB et ETIQC :

a) Comme en 3°, simplifier cette partie du programme en supprimant tout ce qui se rapporte au tableau ESI : on écrira sur la copie le listage simplifié, comme nous l'avons fait dans l'énoncé en 3°.

b) Dire combien de fois joue l'instruction *faire :*

```
pour I depuis 1 pas 1 jusqu'à CARD I faire
    ..................
    ETIQG:fait ;
```

Comme en 3°b et 3°c, on présentera en un tableau les valeurs de MINPENTE et de ISP à la fin de chacune des 9 étapes I=1,..., 9.

c) Quelle est la valeur de CARS à la fin de l'exécution du programme? Marquer sur la figure tous les points I=ISOM[1],..., I = ISOM[CARS], en indiquant à côté de chaque point ISOM[α] un carré $\boxed{\alpha}$.

Quel est en terme géométrique l'effet de la partie du programme qui précède ETIQC ?

1.5 Expliquer par un commentaire général le rôle du tableau ESI :

On mettra de plus une ligne particulière de commentaire pour chaque ligne de l'instruction faire "*pour* I *depuis* 1 *pas* 1 *jusqu'à* CARD I... ... ETIQF : *fait*", où intervient le tableau ESI. Il sera bon d'illustrer ces commentaires d'une figure, tenant compte de l'effet géométrique du programme (trouvé ci-dessus en 4°c).

1.6 On étudie maintenant la partie du programme comprise entre les deux étiquettes ETIQC et FIN :

a) Donner sous forme de tableau les valeurs de AS[H] et BS[H] pour H variant de 1 à CARS.

Pour tout point I = ISOM[H] (H = 1,..., CARS) on considère désormais dans le plan la droite DAP[H] d'équation :

AS[H](X - XI[ISOM[H]]) + BS[H](Y - YI[ISOM[H]]) = 0 ;

ainsi que le demi-plan DEX[H] défini à partir de la droite DAP[H] par l'équation :

AS[H](X - XI[ISOM[H]]) + BS[H](Y - YI[ISOM[H]]) ≤ 0

b) Est-il possible de définir la droite DAP[H] à partir du point ISOM[H] et d'autres points ISOM[H'], ISOM[H"] pour H', H" convenablement choisis? Expliquer pourquoi les cas H = 1 et H = CARS sont traités à part dans le programme.

c) Compte-tenu de l'effet de la première partie du programme (cf 4° c)) dire s'il existe des points I dans le demi-plan DEX[H].

(afin de découvrir les propriétés géométriques recherchées en b et c, on pourra dessiner la droite DAP[H], et hachurer le demi-plan DEX[H], pour une ou plusieurs valeurs de H.

2 <u>Listage de l'algorithme en Algol 68</u> :

```
début
    réel XMI,XMA,MINPENTE,PTIIS,PCORD;
    ent IMI,IMA,I,IS,ISP,CARS,CARDI,H
    [1:CARD I] réel XI,YI,AS,BS;
    [1:CARD I] ent ESI,ISOM

    IMI:=IMA:=1;XMI:=XMA:=XI[1];
    pour I depuis 2 pas 1 jusqu'à CARD I faire
        si XI[I]<s XMI alors XMI:=XI[I];IMI=I
        sinsi XI[I]=XMI et YI[I]<s YI[IMI] alors IMI:=I
        sinsi XI[I]=XMA et YI[I]>s YI[IMA] alors IMA:=I
        sinsi XI[I]>s XMA alors XMA:=XI[I];IMA:=I fsi fait;

commentaire : nous utilisons les signes <s et >s pour exprimer une iné-
galité stricte :

    si XMI=XMA alors
        si YI[IMI]<s YI[IMA] alors
            ISOM[1]:=IMI;ISOM[2]:=IMA;CARS:=2
        sinsi YI[IMI]=YI[IMA] alors ISOM[1]:=IMI;CARS:=1 fsi;
        aller à FIN fsi

    pour I depuis 1 pas 1 jusqu'à CARD I faire ESI[I]:=1 fait;
```

```
    IS:=IMI;CARS:=1;ISOM[1]:=IMI;
    ETIQA:
    PCORD:=(YI[IMA]-YI[IS])/(XI[IMA]-XI[IS]);
    MINPENTE:=PCORD;ISP:=IMA;
    pour I depuis 1 pas 1 jusqu'à CARD I faire
        si ESI[I]=0 alors aller à ETIQF fsi;
        si XI[I]≤XI[IS] alors ESI[IS]:=0; aller à ETIQF fsi;
        PTIIS:=(YI[I]-YI[IS])/(XI[I]-XI[IS]);
        si PTIIS≥PCORD alors ESI[IS]:=0; aller à ETIQF fsi;
        si PTIIS<s MINPENTE alors MINPENTE:=PTIIS;ISP:=I
        sinsi PTIIS=MINPENTE et XI[I]>s XI[ISP] alors ISP:=I
        sinsi PTIIS>s MINPENTE et XI[I]≤XI[ISP] alors ESI[I]:=0 fsi
        ETIQF:fait;

    si XI[ISP]<s XI[IMA]
        CARS:=CARS+1;ISOM[CARS]:=ISP;IS:=ISP; aller à ETIQA fsi;
    si XI[ISP]=XI[IMA] et YI[ISP]<s YI[IMA] alors
        CARS:=CARS+1;ISOM[CARS]:=ISP fsi;
    CARS:=CARS+1;ISOM[CARS]:=IMA;IS:=IMA;

    pour I depuis 1 pas 1 jusqu'à CARD I faire ESI[I]:=1 fait;

    ETIQB:
    PCORD:= YI[IMI]-YI[IS])/(XI[IMI]-XI[IS])
    MINPENTE:=PCORD;ISP:=IMI;
    pour I depuis 1 pas 1 jusqu'à CARD I faire
        si ESI[I]=0 alors aller à ETIQG fsi;
        si XI[I]≥XI[IS] alors ESI[IS]:=0; aller à ETIQG fsi;
        PTIIS:=(YI[I]-YI[IS])/(XI[I]-XI[IS]);
        si PTIIS≥PCORD alors ESI[IS]:=0; aller à ETIQG fsi
        si PTIIS<s MINPENTE alors MINPENTE:=PTIIS;ISP:=I
        sinsi PTIIS=MINPENTE et XI[I]<s XI[ISP] alors ISP:=I
        sinsi PTIIS>s MINPENTE et XI[I]≥XI[ISP] alors ESI[I]:=0 fsi
        ETIQG: fait;

    si XI[ISP]>s XI[IMI] alors
        CARS:=CARS+1;ISOM[CARS]:=ISP; aller à ETIQB
    sinsi XI[ISP]=XI[IMI] et YI[ISP]≥s YI[IMI] alors
        CARS:=CARS+1;ISOM[CARS]:=ISP fsi;

    ETIQC:si CARS≤2 aller à FIN fsi;
    AS[1]:=-YI[ISOM[2]]+YI[ISOM[CARS]];
    BS[1]:= XI[ISOM[2]]-XI[ISOM[CARS]];
    pour H depuis 2 pas 1 jusqu'à CARS-1 faire
        AS[H]:=-YI[ISOM[H+1]]+YI[ISOM[H-1]];
        BS[H]:= XI[ISOM[H+1]]-XI[ISOM[H-1]] fait;
    AS[CARS]:=-YI[ISOM[1]]+YI[ISOM[CARS-1]];
    BS[CARS]:= XI[ISOM[1]]-XI[ISOM[CARS-1]].
    pour H depuis 1 pas 1 jusqu'à CARS faire
        RO:=(AS[H]↑2+BS[H]↑2)↑(1/2);
        AS[H]:=AS[H]/RO;BS[H]:=BS[H]/RO fait
    FIN:
fin
```

3 Solution du problème

3.1 a) Le tableau ci-dessous donne dans la colonne in les valeurs initiales des indices et des abscisses définies par l'instruction : IMI:=IMA:=1 ; XMI:=XMA:=XI[1] ; puis les colonnes suivantes donnent les valeurs obtenues quand on applique l'instruction *faire* en faisant varier I de 2 à 9. Comme le demande l'énoncé on a signalé d'une étoile tout changement de valeur apparaissant dans une colonne relativement à la précédente.

I	in	2	3	4	5	6	7	8	9
IMI	1	2*	2	4*	4	6*	6	8*	8
IMA	1	1	3*	3	5*	5	5	5	9*
XMI	2	2	2	1*	1	1	1	1	1
XMA	2	2	3*	3	4*	4	4	4	4

Tableau 1 : Evolution des indices IMI et IMA et des abscisses XMI et XMA



Figure 1 : Position des 9 points L:{XI[I], YI[I]}

3.1 b) La figure 1 permet de suivre dans le plan la variation des colonnes du tableau 1

3.1 c) Voici le bloc de la question 1°, muni de commentaires.

commentaire : convenons de noter XMI et XMA le minimum et le maximum des valeurs prises par XI[I] sur l'ensemble des CARD I points I :

XMI = inf{XI[I] | I = 1,..., CARD I} ; XMA = sup{XI[I] | I = 1,..., CARD I};

l'objet du bloc est de déterminer XMI et XMA ; simultanément on détermine l'indice IMI du point qui parmi ceux d'abscisse XMI a *la plus faible* ordonnée ; et l'indice IMA du point qui parmi ceux d'abscisse XMA a *la plus forte* ordonnée.

IMI:=IMA:=1 ; XMI:=XMA:=XI[I];

commentaire : on part du point I = 1 ; l'instruction *faire* considère successivement tous les points ;

pour I *depuis* 2 *pas* 1 *jusqu'à* CARD I *faire*
si XI[I]$<_S$XMI *alors* XMI:=XI[I];IMI:=I

commentaire : on compare l'abscisse du point I avec l'abscisse la plus petite déjà rencontrée : si XI[I] est strictement plus petite que XMI, on affecte à ce minimum une nouvelle valeur et on garde en IMI l'indice du point I ;

sinsi XI[I]=XMI *et* YI[I]$<_S$YI[IMI] *alors* IMI:=I

commentaire : si l'abscisse du point I est égale au minimum déjà rencontré, il faut comparer les ordonnées, afin de retenir en IMI, celui des points d'abscisse XMI dont l'ordonnée est la plus faible.

sinsi XI[I]=XMA *et* YI[I]$>_S$YI[IMA] *alors* IMA:=I
sinsi XI[I]$>_S$XMA *alors* XMA:=XI[I];IMA:=I *fsi fait;*

commentaire : de même que l'on compare XI[I] à XMI, on le compare au maximum XMA des abscisses déjà rencontrées ; s'il y a égalité, on retient le point I (IMA:=I) à condition que son ordonnée soit la plus forte parmi celle des points d'abscisse XMA.

3.2 L'instruction objet de la question 2° s'applique si XMI=XMA, c'est-à-dire si tous les points I ont même abscisse ; par exemple on peut avoir :

XI	1	1	1	1
	1	2	3	4
YI	2	4	1	3

I	in	2	3	4
IMI	1	1	3*	3
IMA	1	2*	2	2
XMI	1	1	1	1
XMA	1	1	1	1

Tableaux 2 : à gauche tableau des données ; à droite, évolution des indices IMI et IMA.

Dans ce cas IMI est l'indice du point le plus bas ; et IMA est l'indice du point le plus haut ; CARS = 2 : il y a deux sommets (cf *infra*) le premier est IMI (ISOM[1]=IMI), le second est IMA. On peut même supposer que tous les points I soient confondus en un seul! dans ce cas IMI=IMA=1, et le point I = 1 est sommet unique. Il faut noter que en cas de superposition de plusieurs points le programme s'arrête toujours au premier qu'il recontre (par exemple si XI[I]=IMA et YI[I]= YI[IMA], on garde à IMA sa valeur bien que le nouvel I convienne tout aussi bien...).

3.3a) Le rapport PCORD est la pente du segment joignant les points IS et IMA ; le rapport PTIIS est la pente du segment joignant les points I et IS.

3.3b) En appliquant les instructions du bloc simplifié donné en 1-3° avec la valeur IS = IMI = 8 (cf 3.1a), on trouve les valeurs successives de MINPENTE et de ISP, recensées dans le tableau 3.b : la valeur 7 à laquelle s'arrête ISP donne ISOM[2].

I	in	1	2	3	4	5	6	7	8	9
MINPENTE	2/3	2/3	2/3	0*	0	0	0	-1*	-1	-1
ISP	9	9	9	3*	3	3	3	7*	7	7

Tableau 3.b : Détermination du sommet ISOM[2].

3.3c) A partir de la valeur IS=ISOM[2]=7, on trouve de même ISOM[3]=5 (cf tableau 3.c) puis ISOM[4]=9.

I	in	1	2	3	4	5	6	7	8	9
MINPENTE	3/2	3/2	3/2	1*	1	1	1	1	1	1
ISP	9	9	9	3*	3	5*	5	5	5	5

Tableau 3.c : Détermination du sommet ISOM[3].

Sur le tableau on constate que la valeur ISP=5 succède à la valeur ISP=3, cependant que MINPENTE, pente de (IS,ISP) reste égale à 1 ; ici on a posé l'instruction :

sinsi PTIIS=MINPENTE *et* XI[I]$>_s$XI[ISP] *alors* ISP:=I

en général, cette instruction joue quand sont alignés les trois points IS, I et ISP (7,5,3 dans notre exemple ; cf fig 1) : dans ce cas, si l'abscisse de I est strictement supérieure à celle de ISP, le point I prend la place de ISP (ISP:=I). La figure ci-dessous illustre ces deux cas :



Figure 3.c : à gauche le point I (dernier point introduit dans la boucle) est entre IS et ISP : on ne le retiendra pas ; à droite , I est sur le prolongement de (IS,ISP) au delà de ISP : on pose ISP:=I.

3.4a) Après simplification comme dans 3° cette partie du programme devient :

```
ETIQB:
PCORD:=(YI[IMI]-YI[IS])/(XI[IMI]-XI[IS]);
MINPENTE:=PCORD;ISP:=IMI;
pour I depuis 1 pas 1 jusqu'à CARD I faire
    si XI[I]≥XI[IS] alors aller à ETIQG fsi;
    PTIIS:=(YI[I]-YI[IS])/(XI[I]-XI[IS]);
    si PTIIS≥PCORD alors aller à ETIQG fsi;
    si PTIIS<s MINPENTE alors MINPENTE:=PTIIS;ISP:=I
    sinsi PTIIS=MINPENTE et XI[I]<s XI[ISP] alors ISP:=I fsi;
    ETIQG:fait;
si XI[ISP]>s XI[IMI] alors
    CARS:=CARS+1;ISOM[CARS]:=ISP; aller à ETIQB
sinsi XI[ISP]=XI[IMI] alors
    CARS:=CARS+1;ISOM[CARS]:=ISP fsi;
```

3.4b) L'instruction *pour* du bloc de 3.4a), joue d'abord au moins une fois; elle est ensuite répétée aussi longtemps qu'elle produit une valeur de ISP telle que XI[ISP]$>_s$XI[IMI] ; avec les données considérées ici, l'instruction ne joue qu'une fois, car elle aboutit à ISP=4 ; et XI[ISP]= XI[IMI]=1 ; on a donc ISOM[5]=4 ; CARS=5. Voici comme en 3.3b) et 3.3c)) un tableau expliquant le jeu de l'instruction *pour* :

I	in	1	2	3	4	5	6	7	8	9
MINPENTE	2/3	0^*	0	0	0	0	0	0	0	0
ISP	8	1^*	1	1	4^*	4	4	4	4	4

Tableau 4.b : Détermination du sommet ISOM[5]

3.4c) Le programme détermine les sommets du plus petit polygone convexe contenant tous les points I (polygone de sustentation ; encore appelé enveloppe convexe). Dans une première partie on a obtenu deux de ces sommets, qui sont les points IMI et IMA. (cf 1°). Les autres sommets sont recherchés d'abord en-dessous du segment (IMI,IMA) (cf 3.3) ; et ensuite au dessus de ce segment (cf 3.4). Si l'on convient de dire que le point ISOM[X] est le X-ème sommet, on voit que les sommets sont numérotés à partir du premier qui est IMI (ISOM[1]=IMI) en suivant dans le sens direct le périmètre du convexe. En effet, considérons par exemple comment on trouve les sommets situés en-dessous de (IMI,IMA) partant de IMI=ISOM[1]=IS, on caractérise ISOM[2], comme étant le point I tel que le segment (IMI,I) ait une pente aussi faible que possible ; puis à partir

de ISOM[2]=IS, on caractérise ISOM[3] par les deux conditions d'être situé à droite de ISOM[2] (c'est pourquoi on a l'instruction :
si XI[I]≤XI[IS] *alors aller à* ETIQF *fsi*) et définir avec ISOM[2] un segment (ISOM[2],I) de pente aussi faible que possible ; etc.

3.5 Le tableau ESI est utilisé deux fois : d'une part dans la détermination des sommets situés en dessous de (IMI,IMA) (partie du programme comprise entre ETIQA et ETIQB ; et simplifiée dans la question 3° ; cf § 3.3) ; d'autre part dans la détermination des sommets situés au-dessus de (IMI,IMA) (de ETIQB à ETIQC ; cf question 4° ; § 3.4). Son rôle étant le même dans les deux cas nous nous bornerons à considérer la partie du programme comprise entre ETIQA et ETIQB.

Le rôle du tableau ESI est d'abréger l'exécution de l'instruction *pour* ; car celle-ci commence par :

si ESI[I]=0 *alors aller à* ETIQF *fsi;*

on saute donc les points I pour lesquels ESI[I]=0. Pourquoi ces points sont-ils sautés? parce qu'on est assuré qu'ils ne peuvent pas être des sommets situés en dessous de (IMI,IMA). Au départ le tableau ESI a été mis à 1 par l'instruction :

pour I *depuis* 1 *pas* 1 *jusqu'à* CARD I *faire* ESI[I]:=1 *fait;*

mais progressivement, on élimine des points I, en affectant à ESI[I] la valeur 0.

Plus précisément, l'instruction *pour* comporte trois fois l'instruction ESI[I]:=0 ; pour voir dans chacun de ces trois cas ce qui légitime l'élimination de I, on a marqué sur une figure les zones éliminées (cf figure 5) ; ce qui suggère des commentaires pour les instructions d'élimination.



Figure 5 : On a hachuré pour distinguer les zones éliminées

si XI[I]≤XI[IS] *alors* ESI[IS]:=0 *aller à* ETIQF *fsi;*

commentaire : à partir d'un sommet IS déjà trouvé, on recherche sur le contour apparent le sommet situé sous (IMI,IMA) immédiatement à droite de IS : on élimine donc le demi-plan gauche limité à la droite IS,Y (parallèle à OY menée par IS).

si PTIIS≥PCORD *alors* ESI[IS]:=0 *aller à* ETIQF *fsi;*

commentaire : le sommet cherché (ainsi que les suivants) est certainement situé en dessous de la corde (IS,IMA); on élimine donc le secteur angulaire limité aux demi-droites IS.Y et IS.IMA.

sinsi PTISS$>_S$MINPENTE *et* XI[I]≤XI[ISP] *alors* ESI[I]:= 0 *fsi* *commentaire* : provisoirement on a abouti à un point ISP, qui pourrait être le sommet suivant ISP, car il réalise le minimum de la pente : il est donc certain que le sommet cherché (ni aucun de ceux qui le suivent sur le bord inférieur IMI,IMA) ne peut être situé au dessus de la ligne brisée (IS,ISP,IMA) donc en particulier dans le triangle hachuré.

3.6a) Le tableau 6 donne les valeurs de AS et BS fournies par l'exécution du bloc compris entre les deux étiquettes ETIQC et FIN.

H	1	2	3	4	5
AS	$3/\sqrt{10}$	$-1/\sqrt{10}$	$-3/\sqrt{13}$	$-1/\sqrt{10}$	$2/\sqrt{13}$
BS	$1/\sqrt{10}$	$3/\sqrt{10}$	$2/\sqrt{13}$	$-3/\sqrt{10}$	$-3/\sqrt{13}$

3.6b) La droite DAP[H] n'est autre que la parallèle menée par le sommet ISOM[H] au segment joignant les deux sommets de rang H-1 et H+1 qui précèdent et suivent ISOM[H] sur le contour du polygone convexe. On a dû traiter à part les cas H=1 (pour lequel le sommet précédent a le rang CARS ; et non H-1=0) et H=CARS (pour lequel le sommet suivant a le rang 1 ; et non H+1=CARS+1).



Figure 6 : à gauche, position des droites d'appui DAP[H] et des demi-plans extérieurs DEX[H] ; à droite, explication de l'équation du demi-plan extérieur.

3.6c) Le vecteur {AS[H],BS[H]} est construit en faisant tourner d'un angle droit, dans le sens direct, le vecteur ayant pour origine le point ISOM[H-1] et pour extrémité ISOM[H+1] ; puis en divisant le vecteur obtenu par sa norme. On a ainsi un vecteur unitaire dont les composantes sont les cosinus directeurs de la droite DAP[H]. On voit sur la figure 6 que DAP[H] est une droite d'appui au polygone convexe : i.e. elle touche celui-ci en un sommet, mais ne pénètre pas à l'intérieur : de plus parce que les sommets sont *numérotés dans le sens direct*, le signe des cosinus directeurs est tel que le demi-plan DEX[H] est extérieur au convexe. Pour démontrer cela, posons quelques notations simplifiées :

A : le sommet ISOM[H-1] ; B : le sommet ISOM[H];

C : le sommet ISOM[H+1] ; M : un point (X_M, Y_M).

Ceci posé, le premier membre de l'équation du demi-plan DEX[H] s'écrit :

$$\begin{aligned}\Phi(M) &= AS[H](X_M - X_B) + BS[H](Y_M - Y_B)\\ &= (1/\|AC\|)(-(Y_C - Y_A)(X_M - X_B) + (X_C - X_A)(Y_M - Y_B))\\ &= (1/\|\vec{AC}\|)(\vec{AC} \wedge \vec{BM}).\end{aligned}$$

On doit vérifier que $\Phi(I)$ est positif pour tout point I de l'ensemble donné ; c'est-à-dire qu'est positif $(\vec{AC} \wedge \vec{BI})$; pour cela il suffit de vérifier que $\Phi(A)$ et $\Phi(C)$ sont positifs, puisque tous les vecteurs $\vec{BI}$ sont compris dans l'angle $\vec{BA}$, $\vec{BC}$. Vérifions par exemple que $\Phi(C)$ est positif ; on a :

$$\begin{aligned}\Phi(C) &= (1/\|AC\|)\ (\vec{AC} \wedge \vec{BC})\\ &= (1/\|AC\|)\ ((\vec{AB} + \vec{BC}) \wedge \vec{BC}) = (1/\|AC\|)\ (\vec{AB} \wedge \vec{BC})\ ;\end{aligned}$$

or $\vec{AB} \wedge \vec{BC}$ est bien positif parce que les sommets sont numérotés dans le sens direct.

4 *Le programme CONVAP* :

Comme on l'a expliqué en tête du présent article, le programme CONVAP est une procédure utilisée dans le programme CONVESP, programme qui sert à définir l'enveloppe convexe (convexe polyèdral) d'un ensemble de points de l'espace tridimensionnel.

Outre qu'il est écrit en langage FORTRAN, (dont l'organisation logique ne suit pas exactement celle de l'ALGOL) le programme CONVAP diffère de l'algorithme donné au § 2 en ce que l'on a adjoint une troisième coordonnée ZI en vue d'utiliser CONVAP dans un cadre tridimensionnel. Le programme CONVAP opère en fait uniquement sur les deux coordonnées XI et YI (autrement dit : il construit l'enveloppe convexe du nuage des points en projection sur le plan des X,Y ; ou encore, il construit un contour apparent du convexe spatial) ; mais s'il se trouve qu'un sommet est la projection de plusieurs points de l'espace ayant des ZI différents, c'est toujours le point I le plus haut (i.e. celui dont le ZI est le plus fort) qui est choisi (i.e. dont l'indice I est rangé dans le tableau ISOM) ; de plus on garde dans un tableau DB l'indication DB(I)=1 que ce point est *double* en projection sur le plan des X,Y . On notera que dans l'application du programme CONVESP et de sa procédure CONVAP on suppose que le nuage des points I ne comporte pas de points doubles en trois dimensions (i.e. de points I et IP dont les trois coordonnées XI, YI, ZI soient identiques) ; ce que l'on peut vérifier avant de traiter le tableau de données.

Le listage qui suit est écrit en FORTRAN IV pour l'ordinateur CDC 6600 du CCPN.

```
      SUBROUTINE CONVAP(XI,YI,ZI,SUP,ISUP,NP,CARS,ISOM,ESI,
     1AS,BS,DB)
C     CE SOUS-PROGRAMME RECHERCHE L'ENVELOPPE CONVEXE
C     DANS LE PLAN DEFINI PAR XI ET YI,ET IL CALCULE
C     UNE FORME LINEAIRE D'APPUI AS(H)*XI+BS(H)*YI
C     POUR CHAQUE SOMMET DE CETTE ENVELOPPE.
      REAL XI(NP),YI(NP),ZI(NP),AS(NP),BS(NP),MINPENT
      INTEGER ESI(NP),DB(NP),ISOM(NP),SUP,ISUP,CARS,H
C     ON COMMENCE PAR CHERCHER LE POINT LE PLUS A GAUCHE
C     (IMI D'ABCISSE XMI) ET LE POINT LE PLUS A DROITE
C     (IMA D'ABCISSE XMA)
      IMI=1
      IMA=1
      DB(1)=0
      IF(SUP.NE.1.OR.ISUP.NE.1) GO TO 4
C     LA RECHERCHE DEBUTE PAR LE POINT 1 SAUF SI CELUI-CI
C     EST ELIMINE;DANS CE DERNIER CAS ON DEBUTE EN I=2.
      IMA=2
      IMI=IMA
    4 CONTINUE
      XMA=XI(IMI)
      XMI=XMA
      DO 30 I=2,NP
      IF(SUP.EQ.1.AND.I.EQ.ISUP) GO TO 30
C     SI SUP=1,LE POINT D'INDICE ISUP EST SUPPRIME
      DB(I)=0
      IF(XI(I)-XMI) 5,10,11
    5 XMI=XI(I)
      IMI=I
      GO TO 11
   10 IF(YI(I).NE.YI(IMI)) GO TO 12
      DB(I)=1
      DB(IMI)=1
      IF(ZI(I).GT.ZI(IMI)) IMI=I
C     DB(I)=1 SI LE POINT I EST DOUBLE DANS LE PLAN (X*Y)
C     SI DEUX POINTS SONT SUPPERPPOSES ON RETIENT CELUI
C     POUR LEQUEL Z EST LE PLUS FORT.
   12 IF(YI(I).LT.YI(IMI)) IMI=I
C     A ABCISSE EGALE,ON PREND POUR IMI LE POINT DE
C     PLUS FAIBLE ORDONNEE
   11 CONTINUE
      IF(XI(I)-XMA) 30,15,20
   15 IF(YI(I).NE.YI(IMA)) GO TO 16
      DB(I)=1
      DB(IMA)=1
      IF(ZI(I).GT.ZI(IMA)) IMA=I
   16 IF(YI(I).GT.YI(IMA)) IMA=I
C     A ABCISSE EGALE,ON PREND POUR IMA
C     LE POINT DE PLUS FORTE ORDONNEE.
      GO TO 30
   20 XMA=XI(I)
      IMA=I
   30 CONTINUE
      IF(XMI.NE.XMA) GO TO 45
      IF(YI(IMI)-YI(IMA)) 35,40,45
   35 CONTINUE
      ISOM(1)=IMI
      ISOM(2)=IMA
      CARS=2
      AS(1)=0
      AS(2)=0
      BS(1)=1
      BS(2)=-1
      GO TO 200
```

```
   40  CONTINUE
       ISOM(1)=IMI
       CAPS=1
       AS(1)=0
       BS(1)=1
       GO TO 200
   45  CONTINUE
C      SI XMI=XMA,LE NUAGE EST REDUIT A UN SEGMENT
C      D'ABCISSE XMI,OU PEUT ETRE A UN SEUL POINT;
C      SINON ON DOIT CHERCHER LES SOMMETS AUTRES
C      QUE IMI ET IMA D'ABORD SUR L'ARC INFERIEUR
C      (IMI,IMA) PUIS SUR L'ARC SUPERIEUR.
       DO 50 I=1,NP
       ESI(I)=1
   50  CONTINUE
       IF(SUP.EQ.1) ESI(ISUP)=0
C      ON N'ESSAYE LE POINT I QUE SI ESI(I)=1;
C      UN POINT SUPPRIME N'EST PAS ESSAYE;AU
C      COURS DE LA RECHERCHE DES POINTS IS SONT
C      ELIMINES EN POSANT ESI(I)=0
       IS=IMI
       CARS=1
       ISOM(1)=IMI
   60  CONTINUE
       PCORD=(YI(IMA)-YI(IS))/(XI(IMA)-XI(IS))
       MINPENT =PCORD
       ISP=IMA
       DO 100 I=1,NP
       IF(ESI(I).EQ.0) GO TO 100
       IF(XI(I)-XI(IS)) 65,65,70
   65  ESI(IS)=0.
       GO TO 100
   70  PTIIS=(YI(I)-YI(IS))/(XI(I)-XI(IS))
       IF(PTIIS-PCORD) 80,75,75
   75  ESI(IS)=0
       GO TO 100
   80  CONTINUE
       IF(PTIIS-MINPENT ) 85,90,95
   85  MINPENT =PTIIS
       ISP=I
       GO TO 100
   90  IF(XI(I).NE.XI(ISP)) GO TO 92
       DB(I)=1
       DB(ISP)=1
       IF(ZI(I).LE.ZI(ISP)) GO TO 100
       ISP=I
   92  IF(XI(I).GT.XI(ISP)) ISP=I
       GO TO 100
   95  IF(XI(I).LE.XI(ISP)) ESI(I)=0
  100  CONTINUE
       IF(XI(ISP)-XI(IMA)) 105,110,120
  105  CARS=CARS+1
       ISOM(CARS)=ISP
       IS=ISP
       GO TO 60
  110  IF(YI(ISP).GE.YI(IMA)) GO TO 120
       CARS=CARS+1
       ISOM(CARS)=ISP
  120  CONTINUE
       CARS=CARS+1
       ISOM(CARS)=IMA
       IS=IMA
```

```
C     LA RECHERCHE DES SOMMETS SUR L'ARC INFERIEUR
C     (IMI,IMA) EST TERMINEE;L'ETUDE DE L'ARC
C     SUPERIEUR COMMENCE
      DO 130 I=1,NP
      ESI(I)=1
 130  CONTINUE
      IF(SUP.EQ.1) ESI(ISUP)=0
 140  CONTINUE
      PCORD=(YI(IMI)-YI(IS))/(XI(IMI)-XI(IS))
      MINPENT =PCORD
      ISP=IMI
      DO 180 I=1,NP
      IF(ESI(I).EQ.0) GO TO 180
      IF(XI(I)-XI(IS)) 160,174,174
 174  ESI(IS)=0
      GO TO 180
 160  PTIIS=(YI(I)-YI(IS))/(XI(I)-XI(IS))
      IF(PTIIS-PCORD) 170,165,165
 165  ESI(IS)=0
      GO TO 180
 170  CONTINUE
      IF(PTIIS-MINPENT) 175,176,177
 175  MINPENT=PTIIS
      ISP=I
      GO TO 180
 176  IF(XI(I).NE.XI(ISP)) GO TO 166
      DB(I)=1
      DB(ISP)=1
      IF(ZI(I).LE.ZI(ISP)) GO TO 180
      ISP=I
 166  IF(XI(I).LT.XI(ISP)) ISP=I
      GO TO 180
 177  IF(XI(I).GE.XI(ISP)) ESI(I)=0
 180  CONTINUE
      IF(XI(ISP)-XI(IMI)) 200,185,190
 190  CARS=CARS+1
      ISOM(CARS)=ISP
      IS=ISP
      GO TO 140
 185  IF(YI(ISP)-YI(IMI)) 200,200,195
 195  CONTINUE
      CARS=CARS+1
      ISOM(CARS)=ISP
 200  CONTINUE
C     AS(H) ET BS(H) DESIGNENT LES COSINUS DIRECTEUR
C     DE LA DROITE D'APPUI AU CONTOUR APPARENT EN H.
      NSOM=CARS-1
      AS(1)=-YI(ISOM(2))+YI(ISOM(CARS))
      BS(1)=XI(ISOM(2))-XI(ISOM(CARS))
      DO 210 H=2,NSOM
      J1=H+1
      J2=H-1
      AS(H)=-YI(ISOM(J1))+YI(ISOM(J2))
      BS(H)=XI(ISOM(J1))-XI(ISOM(J2))
 210  CONTINUE
      AS(CARS)=-YI(ISOM(1))+YI(ISOM(NSOM))
      BS(CARS)=XI(ISOM(1))-XI(ISOM(NSOM))
C     NORMALISATION DE AS ET BS
      DO 211 H=1,CARS
      RO=SQRT(AS(H)**2+BS(H)**2)
      AS(H)=AS(H)/RO
      BS(H)=BS(H)/RO
 211  CONTINUE
      RETURN
      END
```